

מצא את המרחק הקצר ביותר בין הישר המחבר את הנקודות

```
p1:=matrix(3,1,[1,2,3]); p2:=matrix(3,1,[3,1,-1])
```

$$\begin{pmatrix} 1 \\ 2 \\ 3 \end{pmatrix}$$

$$\begin{pmatrix} 3 \\ 1 \\ -1 \end{pmatrix}$$

והישר המחבר את הנקודות

```
q1:=matrix(3,1,[1,-1,7]); q2:=matrix(3,1,[1,3,1]);
```

$$\begin{pmatrix} 1 \\ -1 \\ 7 \end{pmatrix}$$

$$\begin{pmatrix} 1 \\ 3 \\ 1 \end{pmatrix}$$

```
l1:=p1+t*(p2-p1);
```

$$\begin{pmatrix} 2t+1 \\ 2-t \\ 3-4t \end{pmatrix}$$

```
l2:=q1+s*(q2-q1)
```

$$\begin{pmatrix} 1 \\ 4s-1 \\ 7-6s \end{pmatrix}$$

```
e1:=linalg::scalarProduct( l1-l2, p2-p1 )
```

$$21t - 20s + 13$$

```
e2:=linalg::scalarProduct( l1-l2, q2-q1 )
```

$$20t - 52s + 36$$

```
solve( {e1=0,e2=0}, {s,t} )
```

$$\left\{ \left[s = \frac{124}{173}, t = \frac{11}{173} \right] \right\}$$

```
l1:=subs(l1,t=11/173)
```

$$\begin{pmatrix} \frac{195}{173} \\ \frac{335}{173} \\ \frac{475}{173} \end{pmatrix}$$


```
l2:=subs(l2,s=124/173)
```

$$\begin{pmatrix} 1 \\ \frac{323}{173} \\ \frac{467}{173} \end{pmatrix}$$

```
norm(l1-l2, 2)
```

$$\frac{2\sqrt{173}}{173}$$

```
plot( plot::Line3d(p1,p2) , plot::Line3d(q1,q2) , plot::Line3d(l1,l2) , plot::Point3d(l1,Color=RGB::Red) ,  
plot::Point3d(l2,Color=RGB::Red) , plot::Point3d(p1,Color=RGB::Green) ,  
plot::Point3d(p2,Color=RGB::Green) , plot::Point3d(q1,Color=RGB::Black) ,  
plot::Point3d(q2,Color=RGB::Black) )
```



מצא את השטח של המשולש הנוצר על ידי שני קווים ישרים היוצאים מהנקודה

```
p:=matrix(3,1,[1,2,3])
```

$$\begin{pmatrix} 1 \\ 2 \\ 3 \end{pmatrix}$$

בכיוונים

```
v1:=matrix(3,1,[1,-3,2]); v2:=matrix(3,1,[2,2,1])
```

$$\begin{pmatrix} 1 \\ -3 \\ 2 \end{pmatrix}$$

$$\begin{pmatrix} 2 \\ 2 \\ 1 \end{pmatrix}$$

והמישור

```
3*x+2*y-z=6
```

$$3x + 2y - z = 6$$

```
l1:=p+t*v1
```

$$\begin{pmatrix} t+1 \\ 2-3t \\ 2t+3 \end{pmatrix}$$

```
linalg::scalarProduct( matrix(3,1,[3,2,-1]), l1) = 6
```

$$4 - 5t = 6$$

```
solve( 4 - 5*t = 6 , t)
```

$$\left\{ -\frac{2}{5} \right\}$$

```
l1:=subs(l1,t=-2/5)
```

$$\begin{pmatrix} \frac{3}{5} \\ \frac{16}{5} \\ \frac{11}{5} \end{pmatrix}$$

```
l2:=p+t*v2
```

$$\begin{pmatrix} 2t+1 \\ 2t+2 \\ t+3 \end{pmatrix}$$

```
linalg::scalarProduct( matrix(3,1,[3,2,-1]), l2) = 6
```

$$9t + 4 = 6$$


```
solve( 9*t+4=6, t)
```

$$\left\{ \frac{2}{9} \right\}$$

```
l2:=subs(l2,t=2/9)
```

$$\begin{pmatrix} \frac{13}{9} \\ \frac{22}{9} \\ \frac{29}{9} \end{pmatrix}$$

```
plot( plot::Line3d(p,l1,Color=RGB::Red), plot::Line3d(p,l2,Color=RGB::Red), plot::Plane(l1,matrix(3,1,[3,2,-1])) )
```



```
A:=1/2*norm( linalg::crossProduct( l1-p,l2-p) , 2 )
```

$$\frac{2\sqrt{122}}{45}$$

מצא את הקו הישר שהוא החיתוך של שני המישורים

```
p1:= 3*x+4*y+5*z=2; p2:=x-2*y+z=8;
```

$$3x + 4y + 5z = 2$$

$$x - 2y + z = 8$$


```
solve( {p1,p2}, {z,y} )
```

$$\left\{ \left[y = \frac{x}{7} - \frac{19}{7}, z = \frac{18}{7} - \frac{5x}{7} \right] \right\}$$

```
l:=matrix(3,1,[x,(x-19)/7,(18-5*x)/7])
```

$$\begin{pmatrix} x \\ \frac{x}{7} - \frac{19}{7} \\ \frac{18}{7} - \frac{5x}{7} \end{pmatrix}$$

```
plot( plot::Plane(matrix(3,1,[2/3,0,0]) , matrix(3,1,[3,4,5])) ,
plot::Plane(matrix(3,1,[8,0,0]) , matrix(3,1,[1,-2,1])) ,
plot::Line3d( subs(l,x=-5), subs(l,x=5) , Color=RGB::Red ) )
```



מצא את הצורות של העקומות הבאות


```
a1:=3*x^2-4*y^2+x+3*y+3=0; a2:=3*x^2-4*y^2+4*x*y+x+3*y+3=0; a3:=3*x^2-4*y^2+16*x*y+x+3*y+3=0;
```

$$3x^2 + x - 4y^2 + 3y + 3 = 0$$


$$3x^2 + 4xy + x - 4y^2 + 3y + 3 = 0$$

$$3x^2 + 16xy + x - 4y^2 + 3y + 3 = 0$$


```
plot( plot::Implicit2d(a1,x=-5..5,y=-5..5) )
```



```
plot( plot::Implicit2d(a2,x=-5..5,y=-5..5) )
```



```
plot( plot::Implicit2d(a3,x=-5..5,y=-5..5) )
```



מצא את הצורות של המשטחים הבאים

```
a1:=x^2+3*y^2+4*z^2+4*x*y+4*x*z-4*y*z+x+2*y+3*z+3=0; a2:=x^2+3*y^2+4*z^2+4*x*y+4*x*z-4*y*z+x+2*y+3*z-6=0;  
a3:=4*x^2+6*y^2+4*z^2+4*x*y+4*x*z-4*y*z+x+2*y+3*z-6=0; a4:=4*x^2+6*y^2+4*z^2+4*x*y+4*x*z-4*y*z+x+2*y+3*z+3=0;
```


$$x^2 + 4xy + 4xz + x + 3y^2 - 4yz + 2y + 4z^2 + 3z + 3 = 0$$

$$x^2 + 4xy + 4xz + x + 3y^2 - 4yz + 2y + 4z^2 + 3z - 6 = 0$$


$$4x^2 + 4xy + 4xz + x + 6y^2 - 4yz + 2y + 4z^2 + 3z - 6 = 0$$

$$4x^2 + 4xy + 4xz + x + 6y^2 - 4yz + 2y + 4z^2 + 3z + 3 = 0$$


```
plot( plot::Implicit3d(a1,x=-5..5,y=-5..5,z=-5..5, Mesh=[20,20,20]) )
```



```
plot( plot::Implicit3d(a2,x=-5..5,y=-5..5,z=-5..5, Mesh=[20,20,20]) )
```



```
plot( plot::Implicit3d(a3,x=-5..5,y=-5..5,z=-5..5, Mesh=[20,20,20]) )
```



```
plot( plot::Implicit3d(a4,x=-5..5,y=-5..5,z=-5..5) )
```

